



جامعة الزقازيق – كلية الهندسة – قسم هندسة الحاسبات والمنظومات



Microsoft  
**Visual Basic 6.0**

# CSE100 الحاسبات والبرمجة ٢

د/ عمرو أحمد زامل

[https://dramrzamel.github.io/CSE100\\_VB6/](https://dramrzamel.github.io/CSE100_VB6/)

or

[bit.ly/CSEVB6](https://bit.ly/CSEVB6)

المحاضرة ١٠ : الروسومات



Copyright ©2017 Dr.EngAmr Zamel as part of the course work and learning material. All Rights Reserved.

Where otherwise noted, this work is licensed under a [Creative Commons Attribution-NonCommercial-ShareAlike 4.0 International License](https://creativecommons.org/licenses/by-nc-sa/4.0/).

## في المحاضرة السابقة

■ التعامل مع الرسومات و الصور:

■ المتحكمات البيانية: الخطوط والأشكال. Line & Shape controls

■ متحكمات الصور و الرسومات. Image & PictureBox controls

■ استخدام صندوق قائمة الملفات File-List box

- صناديق التعامل مع الرسومات و الصور:
  - طرق الرسم Graphics methods:
  - الخطوط، الدوائر، طبع النصوص، توقيع نقطة،...
  - نسخ الصور والأشكال مع امكانية تكبير جزء من صورة.
- الملخص .

## طرق الرسم Graphics methods:

■ أهم الطرق لانتاج رسومات في الـ VB:

Method	وظيفتها
Line	رسم خط أو صندوق (مستطيل، مربع، ..)
Circle	رسم دائرة أو شكل بيضاوي
PSet	وضع أو توقيع نقطة على كيان ما
Point	ارجاع لون نقطة معينة
PaintPicture	نسخ صورة من كيان على كيان آخر
Print	اضافة أو كتابة نص على كيان
Cls	مسح منطقة الخرج على كيان ما

■ أهم الكيانات التي يمكن الرسم/الكتابة عليها: النموذج، صندوق الرسومات +

كيان الطابعة Printer

## طرق الرسم Graphics methods:

ملاحظة هامة جدا:

- عند الرسم على كيان ما فإن شكل ولون الرسم الناتج يتحدد بناء على بعض خصائص الكيان الذي نرسم عليه. وأهم هذه الخصائص هي:

الخاصية	التاثير
FillColor	اللون المستخدم لملء الشكل
Draw Width	سمك الخط
FillStyle	نمط أو طريقة ملئ الشكل (خطوط مائلة، متقطعة،...)
ForeColor	لون الخط الخارجى للرسم

- كالعادة اذا لم يتم ذكر الكيان الذي نرسم عليه فان **النموذج** هو الكيان الحالي للرسم.
- هناك خاصيتين مرتبطتين بالرسم هما **CurrentX** و **CurrentY** و دائما تشيران لأخر نقطة تم رسمها علي الكيان صاحبهما.

Object.Line (x1,y1) - (x2,y2) [,Par1, Par2]

- حيث أنه يلزم كتابة احداثيات نقطة البداية **x1,y1** وكذلك نقطة النهاية **x2,y2**
- المعامل **Par1** اختياري ولكن يمكن تحديد لون الرسم فيه **vbRed, vbBlue, vbGreen**.
- المعامل **Par2** اختياري أيضا ويمكن أن نكتب فيه **B** وبالتالي يتم رسم صندوق، أو **BF** ليتم ملء الصندوق المرسوم باللون المحدد في خاصية **ForeColor** للكيان **Object**

## طريقة الخطوط Line:

Object.Line (x1,y1) - (x2,y2) [,Par1, Par2]

■ يمكن تغيير خاصيتي **FillColor** , **ForeColor** لأي كيان في وضع التشغيل و عليه

يمكن تحديد اللون المطلوب مثل **vbRed**, **vbBlue**, **vbGreen**,...


■ اذا لم يتم تحديد نقطة البداية فانها تكون Current position أي نقطة النهاية

لآخر خط أو شكل تم رسمه. وتشير اليه خاصيتي

**CurrentY** و **CurrentX**


## طريقة الخطوط :Line

```
Private Sub Form_Click()  
Dim xSW As Single, ySH As Single  
xSW = Form1.ScaleWidth  
ySH = Form1.ScaleHeight  
Form1.DrawWidth = 3  
Line (0, 0)-(xSW, ySH), vbRed  
End Sub
```



مثلث متساوي الساقين


```
Private Sub Form_Click()  
Line (500, 500)-(2000, 500), vbRed  
Line -(500, 2000), vbBlue  
Line -(500, 500)  
End Sub
```







## طريقة الخطوط Line:

```
Private Sub Form_Click()
Line (1000, 750)-(2400, 1550), vbRed, BF
End Sub
```



رسم مربع علي متحكم صندوق رسومات


```
Private Sub Form_Click()
Picture1.Line (0, 0)-(1000, 1000), , BF
End Sub
```



## طريقة الخطوط Line:

- من المعلوم ان وحدة قياس الأبعاد الغيائية هي ال **Twip** وبالتالي جميع الأبعاد المستخدمة في انتاج رسومات في الأمثلة السابقة تكون بال **Twip** ولكن يمكن تغيير وحدة القياس من خلال


خاصية **ScaleMode**



يمكن أيضا تغيير هذه الخاصية  
للكيان في وضع التشغيل ولكن في  
هذه الحالة سوف نستخدم القيم  
العددية.


## طريقة الخطوط Line:

```
Private Sub Form_Click()  
Picture1.ScaleMode = 7  
Picture1.Line (0, 0)-(2, 2), vbBlue, BF  
End Sub
```



مربع طول ضلعه 2 سم على متحكم صندوق  
رسومات


```
Private Sub Form_Click()  
Line (0.5, 0.75)-(2, 1.75), , B  
End Sub
```



مستطيل أبعاده 1.5 x 1 بوصة علماً بأن وحدة القياس  
للنموذج مضبوط على Inch في وضع التصميم.


رسم مثلث متساوي الأضلاع و طول ضلعه 5 سم.

```
Private Sub Form_Click()
Dim x1 As Single, y1 As Single
Dim x2 As Single, y2 As Single
Dim x3 As Single, y3 As Single
Dim fTL As Single, fTH As Single
ScaleMode = 7
fTL = 5
fTH = fTL * Sqr(3) / 2
x1 = 4
y1 = 0.5
x2 = x1 - fTL / 2
x3 = x1 + fTL / 2
y2 = y1 + fTH
y3 = y2
Line (x1, y1)-(x2, y2)
Line -(x3, y3)
Line -(x1, y1)
End Sub
```




وضح كيف يمكن رسم شكل سداسي منتظم في منتصف النموذج  
تماما بحيث يكون طول ضلعه 1000 تويب وذلك في وضع  
التشغيل علما بان ابعاد النموذج هي  $\text{Height} = 3000, \text{Width} = 4000$

يمكن رسم شكل سداسي بأكثر من طريقة وأحد أسهل هذه الطرق هو اعتبار أن الشكل السداسي يمكن أن يقسم إلى مثلثات متساوية الأضلاع:



## حل مثال 2:

```
Private Sub Form_Click()  
DrawWidth = 2  
Dim fHL, fTH As Single  
Dim xc As Single, yc As Single, x1 As Single, y1 As Single  
Dim x2 As Single, y2 As Single, x3 As Single, y3 As Single  
Dim x4 As Single, y4 As Single, x5 As Single, y5 As Single  
Dim x6 As Single, y6 As Single  
fHL = 1000  
fTH = fHL * Sqr(3) / 2  
xc = Form1.ScaleWidth / 2  
yc = Form1.ScaleHeight / 2  
x1 = xc - fHL / 2  
y1 = yc - fTH  
x2 = x1 + fHL  
y2 = y1  
x3 = xc + fHL  
y3 = yc  
x4 = x2  
y4 = yc + fTH  
x5 = x1  
y5 = y4  
x6 = xc - fHL  
y6 = y3  
Line (x1, y1)-(x2, y2)  
Line -(x3, y3)  
Line -(x4, y4)  
Line -(x5, y5)  
Line -(x6, y6)  
Line -(x1, y1)  
End Sub
```




Object.Circle (x,y), radius [,color, start, end]

- **(x,y)** تمثل احداثيات مركز الدائرة.
- **radius** تمثل نصف القطر.
- **color** هي معامل (قيمة) اختيارية وتمثل لون الخط المرسوم به الدائرة.
- **start & end**: (معاملات اختيارية) يمثلان زوايا قطرية ( $-2\pi : 2\pi$ ) لبداية و نهاية جزء أو قطاع **Sector** من الدائرة، و اذا كان قيمة أي منهما سالبة فانه يتم رسم خط من مركز الدائرة الي محيطها.
- لاحظ أن خصائص الكيان المرسوم عليه الدائرة تؤثر على شكلها مثل **FillColor** , **FillStyle**.
- عند رسم دائرة فان خاصيتي **CurrentX** و **CurrentY** دائما تشير الي مركز الدائرة.



## الدوائر Circles:


```
Private Sub Form_Click()  
FillStyle = 6  
FillColor = vbRed  
DrawWidth = 3  
Circle (1000, 1000), 500, vbGreen  
End Sub
```




```
FillStyle = 6  
FillColor = vbRed  
DrawWidth = 3  
Const pi = 22 / 7  
Circle (2000, 2000), 1000, vbRed, 0, -pi / 2
```



```
FillStyle = 6  
FillColor = vbRed  
DrawWidth = 3  
Const pi = 22 / 7  
Circle (2000, 2000), 1000, vbRed, -pi / 2, -3 * pi / 2
```




```
Private Sub Form_Click()  
FillStyle = 6  
FillColor = vbRed  
DrawWidth = 3  
Const pi = 22 / 7  
Circle (2000, 2000), 1000, vbRed, -3 * pi / 2, -0.0000001  
Circle (2000, 2000), 1000, vbRed, -pi / 2, -pi  
Circle (2000, 2000), 1000, vbGreen, 0, pi / 2  
Circle (2000, 2000), 1000, vbBlue, pi, 3 * pi / 2  
  
End Sub
```



```
Private Sub Form_Click
FillStyle = 0
FillColor = vbRed
DrawWidth = 3
Const pi = 22 / 7
Circle (2000, 2000), 1000, vbRed, -0.000001, -pi / 2
FillColor = vbGreen
Circle (2000, 2000), 1000, vbGreen, -pi / 2, -pi
FillColor = vbBlue
Circle (2000, 2000), 1000, vbBlue, -pi, -3 * pi / 2
FillColor = vbYellow
Circle (2000, 2000), 1000, vbYellow, -3 * pi / 2, -0.000001

End Sub
```




وضح كيف يمكن رسم مثلث متساوي الأضلاع على النموذج بقاعدة أفقية تمتد بين (2000,3000) و (1000,3000) ويراد رسم دائرة داخل المثلث مقسمة الى أربع أجزاء ملونة بالأحمر والأخضر والأزرق والأصفر.

```

Private Sub Form_Click()
DrawWidth = 2
Const pi = 22 / 7
Dim xc As Single, yc As Single, fradius As Single
Dim x3 As Single, y3 As Single
Dim fTH As Single
fTH = 1000 * Sqr(3) / 2
'or you can use also
'fTH = Sqr(1000 ^ 2 - 500 ^ 2)
x3 = 1500
'draw the triangle
y3 = 3000 + fTH
Line (1000, 3000)-(2000, 3000)
Line -(x3, y3)
Line -(1000, 3000)
'draw the circle
xc = 1500
yc = 3000 + fTH / 3
fradius = fTH / 3
FillStyle = 0
FillColor = vbRed
Circle (xc, yc), fradius, vbRed, -0.0000001, -pi / 2
FillColor = vbBlue
Circle (xc, yc), fradius, vbBlue, -pi / 2, -pi
FillColor = vbGreen
Circle (xc, yc), fradius, vbGreen, -pi, -3 * pi / 2
FillColor = vbYellow
Circle (xc, yc), fradius, vbYellow, -3 * pi / 2, -0.00000001
End Sub

```



Object.PSet(x,y)

- يتم رسم النقطة علي الكيان **Object** عند الاحداثيات **x,y**
- لون النقطة يتحدد من خاصية **ForeColor** للكيان.
- حجم النقطة يتحدد من خاصية **DrawWidth** للكيان وتبدأ من قيمة ١
- يمكن استخدام هذه الطريقة في رسم الدوال الهندسية
- يمكن توقيع نقاط بطريقة عشوائية باستخدام الدالة **Rnd**
- **طريقة Print**: لاضافة نص عند نقطة الرسم الحالية **CurrentX** و **CurrentY** للكيان **Object**


Object.Print "text"

■ الدالة **Rnd** تقوم بتوليد قيمة عشوائية بين **0** و **1** ولكن يمكنها توليد قيمة عشوائية بين **0** و قيمة أخرى **MaxValue** كالتالي:

$$x = \text{MaxValue} * \text{Rnd}$$

توقيع 50 نقطة عشوائية علي النموذج

```
Private Sub Form_Click()  
    Dim x, y As Integer  
    Dim i As Integer  
  
    Form1.DrawWidth = 5  
  
    For i = 0 To 50  
        x = Form1.ScaleWidth * Rnd  
        y = Form1.ScaleHeight * Rnd  
        PSet (x, y)  
    Next i  
  
End Sub
```



وضح كيف يمكن رسم الدالة  $y = x^2$  على النموذج

حيث أن  $0 \leq x \leq 1$  و  $x$  تتغير كل 0.1


ارسم أيضا المحور الأفقي والرأسي واستخدام مقياس رسم

مناسب للاتجاه الأفقي والرأسي



## حل مثال 4:

```
Private Sub Command1_Click()  
Dim xfun As Single, yfun As Single  
Dim x As Single, y As Single  
CurrentX = 400  
CurrentY = 400  
Print "y"  
Line (500, 500)-(500, 3000)  
Line -(3000, 3000)  
Print "x"  
DrawWidth = 2  
xfun = 0  
For i = 1 To 10  
    xfun = xfun + 0.1  
    yfun = xfun ^ 2  
    x = 500 + 1000 * xfun  
    y = 3000 - 2000 * yfun  
    PSet (x, y)  
Next i  
ForeColor = vbRed  
Print "y=x^2"  
End Sub
```




تم استخدام مقياس رسم للاتجاه الأفقي  
حيث انه كل 1000 تويب تقابل وحدة قياس  
لل  $x$ . وكذلك كل 2000 تويب تقابل وحدة  
قياس لل  $y$ .

## حل مثال 4:

```
Private Sub Command1_Click()  
Dim xfun As Single, yfun As Single  
Dim x As Single, y As Single  
CurrentX = 400  
CurrentY = 400  
Print "y"  
Line (500, 500)-(500, 3000)  
Line -(3000, 3000)  
Print "x"  
DrawWidth = 2  
xfun = 0  
CurrentX = 500  
CurrentY = 3000  
For i = 1 To 10  
    xfun = xfun + 0.1  
    yfun = xfun ^ 2  
    x = 500 + 1000 * xfun  
    y = 3000 - 2000 * yfun  
    Line -(x, y)  
Next i  
ForeColor = vbRed  
Print "y=x^2"  
End Sub
```

إذا تم استبدال  $\text{Pset}(x,y)$  بـ  $\text{Line}-(x,y)$

فان الرسم يكون متصل



وضح كيف يمكن رسم دائرة باللون الأخضر في مركز النموذج بنصف قطر 750 وعند نقر نقطة بالفأرة على النموذج يتم تحديد اذا ما كانت هذه النقطة تقع داخل الدائرة أولا و اخراج احداثياتها علي صندوق رسائل.





- لابد من استخدام طريقة **Point(x,y)** التي ترجع لون النقطة ذات الاحداثيات **x,y**
- كذلك الحدث الذي سيمكننا من تحديد مكان النقر بالفأرة على النموذج هو حدث

**Form\_MouseDown()**

```
Private Sub Command1_Click()  
Dim X As Single, Y As Single  
Dim xc As Single, yc As Single  
xc = Form1.ScaleWidth / 2  
yc = Form1.ScaleHeight / 2  
FillStyle = 0  
FillColor = vbGreen  
Circle (xc, yc), 750, vbGreen  
End Sub
```

```
Private Sub Form_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)  
If Point(X, Y) = vbGreen Then  
    MsgBox "The point ( " & X & " , " & Y & " ) is in the circle"  
Else  
    MsgBox "The point is not in the circle"  
End If  
End Sub
```

## حل مثال 5: وضع التنفيذ



## طريقة مسح الشاشة Cls:

تقوم Cls بمسح جميع الرسومات الموجودة على النموذج أو صندوق الرسومات.

## طريقة نسخ الصور والأشكال PaintPicture:

**Dst.PaintPicture** Src.Picture Xd, Yd, Wd, Hd, Xs, Ys, Ws, Hs

الكيان الذي يتم نسخ الصورة عليه

الكيان الذي ننسخ الصورة منه

احداثي بداية نسخ الصورة على Dst

عرض وارتفاع جزء الصورة على Dst

احداثي بداية نسخ الصورة من Src

عرض وارتفاع جزء الصورة المطلوب من Src

إذا لم يتم ذكرهم فإنه يتم  
نسخ الصورة كاملة

# طريقة نسخ الصور واسكال :PaintPicture

```
Private Sub Form_Click()  
Picture1.PaintPicture Form1.Picture, 0, 0, 2000, 2000, 1000, 750, 750, 750  
End Sub
```



١. وضح كيف يمكن رسم الأشكال التالية:

■ مثلث متساوي الأضلاع داخل دائرة.

■ مربع داخل دائرة.

■ شكل سداسي منتظم داخل مربع.

٢. وضح كيف يمكن رسم الدالة التالية مع رسم محور أفقي ورأسي:

$$y = \begin{cases} x * 0.1 * \exp(x), & \text{if } 0 \leq x \leq 2, \\ 2, & \text{if } 2 \leq x \leq 4, \\ 6 - x, & 4 \leq x \leq 6. \end{cases}$$

علما بأن السنتيمتر هو وحدة القياس المطلوبة واعتبر أن  $x$  تتغير كل 0.1 في جميع الفترات.